CHAPTER 11

Ordinary Differential Equations

11.1 ANALYTICAL SOLUTIONS

Simply put, a differential equation is an equation expressing a relationship between a function and one or more of its derivatives. A function which satisfies a differential equation is called a *solution*.

The *Mathematica* command **DSolve** is used to solve differential equations. As with algebraic or transcendental equations, a double equal sign, **==**, is used to separate the two sides of the equation.

- **DSolve[*equation*, y[x], x]** gives the general solution, y[x], of the differential equation, *equation*, whose independent variable is x.
- **DSolve[*equation*, y, x]** gives the general solution, y, of the differential equation expressed as a "pure" function (see Appendix) within a list. ReplaceAll (/.) may then be used to evaluate the solution. Alternatively, one may use Part or [[ ]] to extract the solution from the list.

EXAMPLE 1

To solve the first-order differential equation $\frac{dy}{dx} = x + y$, we simply type

```
DSolve[y'[x] == x + y[x], y[x], x]
{{y[x] → - 1 - x + e^x C[1]}}
```

EXAMPLE 2

To obtain the solution of $\frac{dy}{dx} = x + y$ as a pure function, we enter

```
solution = DSolve[y'[x] == x + y[x], y, x]
{{y → (- 1 + e^#1 C[1] - #1 &)}}
```

If we wish to evaluate the solution, we can type

```
y[x] /. solution
{-1 - x + e^x C[1]}
```

Using the pure function, we can evaluate the derivatives of the solution. This would be clumsy using the solution of Example 1.

```
y'[x] /.solution
{-1 + e^x C[1]}
y''[x] + y'[x] /. solution
{-1 + 2 e^x C[1]}
```

We can define a function `f` representing the solution

```
f = solution[[1, 1, 2]]
-1 + e^#1 C[1] - #1 &
```

We can then directly evaluate `f` or any of its derivatives.

```
f[x]
-1 - x + e^x C[1]
f'[x]
-1 + e^x C[1]
f''[x]
e^x C[1]
```

It is *extremely important* that the unknown function be represented `y[x]`, not `y`, within the differential equation. Similarly, its derivatives must be represented `y'[x]`, `y''[x]`, etc. The next example illustrates some common errors.

EXAMPLE 3

```
DSolve[y'[x] == x + y, y[x], x]
```

$$\left\{\left\{y[x] \to \frac{x^2}{2} + x\,y + C[1]\right\}\right\}$$

← *Mathematica* treats y as a constant.

```
DSolve[y' == x + y, y, x]
```

← The function and its derivative must be specified as `y[x]` and `y'[x]`.

```
DSolve::nvld:
 The description of the equations appears to be ambiguous or invalid.
DSolve::deqx :
 Supplied equations are not differential equations of the given functions.
```

The solution of a first order differential equation *without* initial conditions involves an arbitrary constant labeled, by default, `C[1]`. Additional constants (for higher-order equations) are labeled `C[2]`, `C[3]`, ... If a different labeling is desired, the option **DSolveConstants** may be used.

- **DSolveConstants → constantlabel** specifies that the constants should be labeled `constantlabel[1]`, `constantlabel[2]`, etc.

EXAMPLE 4

```
DSolve[y'[x] == x + y[x], y[x], x, DSolveConstants → mylabel]
{{y[x] → -1 - x + e^x mylabel[1]}}
```

Higher-order differential equations are solved in a similar manner. The derivatives are represented `y'[x]`, `y''[x]`, `y'''[x]`, ... Alternatively, `D`, ∂, or `Derivative` may be used.

EXAMPLE 5

```
DSolve[y''[x] + y[x] == 0, y[x], x]
{{y[x] → C[2] Cos[x] - C[1] Sin[x]}}
DSolve[D[y[x], {x, 2}] + y[x] == 0, y[x], x]
{{y[x] → C[2] Cos[x] - C[1] Sin[x]}}
DSolve[∂{x,2} y[x] + y[x] == 0, y[x], x]
{{y[x] → C[2] Cos[x] - C[1] Sin[x]}}
DSolve[Derivative[2][y][x] + y[x] == 0, y[x], x]
{{y[x] → C[2] Cos[x] - C[1] Sin[x]}}
```

Complex differential equations are solved, if possible, using special functions. If *Mathematica* cannot solve the equation, it will either return the equation unsolved, or in terms of unevaluated integrals. In such cases a numerical solution (see Section 11.2) may be more appropriate.

EXAMPLE 6

$x^2\dfrac{d^2y}{dx^2} + x\dfrac{dy}{dx} + (x^2-4)y = 0$ is a special case of Bessel's equation. The solution is expressed in terms of Bessel functions of the first (`BesselJ`) and second (`BesselY`) kind.

```
DSolve[x² y''[x] + x y'[x] + (x² - 4) y[x] == 0, y[x], x]
{{y[x] → BesselJ[2, x] C[1] + BesselY[2, x] C[2]}}
```

EXAMPLE 7

$\dfrac{d^2y}{dx^2} + \dfrac{dy}{dx} + y^2 = 0$ is a nonlinear differential equation which *Mathematica* cannot solve.

```
DSolve[y''[x] + y'[x] + y[x]² == 0, y[x], x]
DSolve[[y[x]² + y'[x] + y''[x] == 0, y[x], x]
```

If values of y, and perhaps one or more of its derivatives, are specified along with the differential equation, the task of finding y is known as an *initial value problem*. The differential equation and the initial conditions are specified as a list within the `DSolve` command. A unique solution is returned, provided an appropriate number of initial conditions are supplied.

EXAMPLE 8

Solve the equation $\dfrac{dy}{dx} = x + y$ with initial condition $y(0) = 2$. Then plot the solution.

```
equation = DSolve[{y'[x] == x + y[x], y[0] == 2}, y[x], x]
{{y[x] → -1 + 3 e^x - x}}
Plot[y[x] /. equation, {x, -5, 2}];
```



Here is another way the solution can be plotted:

```
solution = DSolve[{y'[x] == x + y[x], y[0] == 2}, y, x]
{{y → ( - 1 + 3 e^#1 - #1 &)}}
f = solution[[1, 1, 2]]
- 1 + 3 e^#1 - #1 &
Plot[f[x],{x, - 5, 2}];
```



A useful way of visualizing the solution of a first-order differential equation is to introduce the concept of a vector field. A vector field on $\mathbb{R}^2$ is a function **F** that assigns to each point (x, y) a two-dimensional vector $\mathbf{F}(x, y)$. By drawing the vectors $\mathbf{F}(x, y)$ for a (finite) subset of $\mathbb{R}^2$, one obtains a geometric interpretation of the behavior of **F**.

- **PlotVectorField[{Fx, Fy}, {x, xmin, xmax}, {y, ymin, ymax}]** produces a vector field plot of the two-dimensional vector function **F**, whose components are **Fx** and **Fy**. The direction of the arrow is the direction of the vector field at the point (x, y). The magnitude of the arrow is proportional to the magnitude of the vector field.

PlotVectorField is contained within the package **Graphics`PlotField`** which must be loaded prior to its use.

EXAMPLE 9

Plot the vector field $\mathbf{F}(x, y) = -y\mathbf{i} + x\mathbf{j}$. By default, no axes are drawn so the option **Axes → Automatic** will be used.

```
<<Graphics`PlotField`
PlotVectorField[{ - y, x}, {x, - 5, 5}, {y, - 5, 5}, Axes → Automatic];
```



Any first-order differential equation can be used to define a vector field. Indeed, the vector field $\mathbf{i} + f(x, y)\mathbf{j}$, corresponding to the equation $\frac{dy}{dx} = f(x, y)$, generates a field whose vectors are tangent to the solution at any point. The next example, although simple, illustrates this nicely.

EXAMPLE 10

Plot the vector field of the solution of the equation $\frac{dy}{dx} = 2x$. The solutions to this equation, parabolas $y = x^2 + c$, can be seen quite vividly.

```
<<Graphics`PlotField`
PlotVectorField[{1, 2x}, {x, -1, 1}, {y, -1, 1}, Axes→Automatic];
```



EXAMPLE 11

In this example we plot the vector field generated by the equation $\frac{dy}{dx} = 2x + y$. The option **HeadLength → 0** suppresses the vector heads for a clearer picture. Then the solutions with initial conditions $y(0) = -2, -1, 0, 1,$ and 2 are plotted on the vector field for comparison.

```
<< Graphics`PlotField`
vf = PlotVectorField[{1, 2x + y}, {x, -2, 1}, {y, -4, 6},
   Axes → Automatic, HeadLength → 0,
   AspectRatio → 1/GoldenRatio,
   DisplayFunction → $DisplayFunction];
```



```
sols = Table[DSolve[{y'[x] == 2x + y[x], y[0] == k}, y[x], x], {k, - 2, 2}]
{{{y[x] → - 2 - 2 x}}, {{y[x] → - 2 + ℯ^x - 2 x}}, {{y[x] → - 2 + 2 ℯ^x - 2 x}},
 {{y[x] → - 2 + 3 ℯ^x - 2 x}}, {{y[x] → - 2 + 4 ℯ^x - 2 x}}}
Do[g[k] =
 Plot[sols[[k, 1, 1, 2]], {x, - 2, 1}, PlotRange → All,
 DisplayFunction → Identity], {k, 1, 5}]
Show[g[1], g[2], g[3], g[4], g[5], vf, DisplayFunction → $DisplayFunction];
```



A *system* of differential equations consists of n differential equations involving $n+1$ variables. Solving a system of differential equations with *Mathematica* is similar to solving a single equation. The next example illustrates how to solve the system $\frac{dx}{dt} = t^2$, $\frac{dy}{dt} = t^3$ with initial conditions $x(0) = 2$, $y(0) = 3$.

EXAMPLE 12

```
solution = DSolve[{x'[t] == t², y'[t] == t³, x[0] == 2, y[0] == 3}, {x[t], y[t]}, t]
```

$$\left\{\left\{x[t] \to \frac{1}{3}(6 + t^3),\ y[t] \to \frac{1}{4}(12 + t^4)\right\}\right\}$$

Instead of specifying the values of f and its derivatives at a single point, values at two distinct points may be given. The problem of solving the differential equation then becomes known as a *boundary value problem*. However, unlike initial value problems, which can be shown to have unique solutions for a wide variety of cases, a boundary value problem may have no solution even for the simplest of equations.

EXAMPLE 13

Consider the equation $\frac{d^2y}{dx^2} + y = 0$ with boundary conditions $y(0) = 0$, $y(\pi) = 1$.

```
DSolve[{y''[x] + y[x] == 0, y[0] == 0, y[π] == 1}, y[x], x]
{}
```

The same equation with $y(0) = 0$, $y(\pi/2) = 1$ has a unique solution.

```
DSolve[{y''[x] + y[x] == 0, y[0] == 0, y[π/2] == 1}, y[x], x]
{{y[x] → Sin[x]}}
```

SOLVED PROBLEMS

11.1 Solve the differential equation $\frac{dy}{dx} = xy$ with initial condition $y(1) = 2$ and graph the solution for $-2 \le x \le 2$.

SOLUTION

```
equation = DSolve[{y'[x] == x y[x], y[1] == 2}, y[x], x]
```

$\left\{\left\{y[x] \to 2\, e^{-\frac{1}{2}+\frac{x^2}{2}}\right\}\right\}$

```
Plot[y[x] /. equation, {x, -2, 2}];
```



11.2 Plot the vector field for the differential equation of the previous example.

SOLUTION

```
<< Graphics`PlotField`
PlotVectorField[{1, x y}, {x, -2, 2}, {y, -10, 10},
                AspectRatio → 1/GoldenRatio, ScaleFactor → 2,
                HeadLength → 0.01, Axes → Automatic];
```



> `ScaleFactor` → ***length*** is an option that scales the length of the vectors so that the longest vector is of length *length*. The default is `Automatic`. At this setting, the vectors often appear too short to view their direction clearly. `Headlength` adjusts the size of the arrowheads.

11.3 Plot the vector field for the equation $\dfrac{dy}{dx} = x^2 + y$ together with its solutions for $y(0) = 0, 1, 2, 3,$ and 4.

SOLUTION

```
<< Graphics`PlotField`
vf = PlotVectorField[{1, x² + y}, {x, 0, 1}, {y, 0, 12},
     Axes → Automatic, HeadLength → 0,
     ScaleFactor → 0.25, AspectRatio → 1/GoldenRatio,
     DisplayFunction → Identity];
sols = Table[DSolve[{y'[x] == x² + y[x], y[0] == k}, y[x], x], {k, 0, 4}];
Do[g[k] = Plot[sols[[k, 1, 1, 2]], {x, 0, 1}, PlotRange → All,
     DisplayFunction → Identity], {k, 1, 5}]
Show[g[1], g[2], g[3], g[4], g[5], vf, DisplayFunction → $DisplayFunction];
```



11.4 The *escape velocity* is the minimum velocity with which an object must be propelled in order to escape the gravitational field of a celestial body. Compute the escape velocity for the planet Earth.

SOLUTION

We shall assume that the initial velocity is in a radial direction away from the Earth's center. According to Newton's laws of motion, the acceleration of a particle is inversely proportional to the square of the distance of the particle from the center of the Earth. If r represents that distance, R the radius of the Earth (approximately 3,960 miles), v the velocity of the particle, and a its acceleration, then $a = \dfrac{dv}{dt} = \dfrac{k}{r^2}$. At the Earth's surface ($r = R$), $a = -g$, where $g = 32.16$ ft/sec^2 = .00609 mi/sec^2. It follows that $k = -gR^2$ so $a = -\dfrac{gR^2}{r^2}$. Since $a = \dfrac{dv}{dt}$ and $v = \dfrac{dr}{dt}$, by the chain rule we have $a = \dfrac{dv}{dt} = \dfrac{dv}{dr}\dfrac{dr}{dt} = v\dfrac{dv}{dr}$. If v_0 represents the escape velocity, we are led to the differential equation $v\dfrac{dv}{dr} = -\dfrac{gR^2}{r^2}$ with initial condition $v = v_0$ when $r = R$.

```
DSolve[{v[r]v'[r] == - g R²/r², v[R] == v₀}, v[r], r]
```

$$\left\{\left\{v[r] \to -\sqrt{-2\,g\,R + \frac{2\,g\,R^2}{r} + v_0^2}\right\}, \left\{v[r] \to \sqrt{-2\,g\,R + \frac{2\,g\,R^2}{r} + v_0^2}\right\}\right\}$$

Since the velocity is positive at the surface of the Earth ($r = R$), and must remain positive for the duration of the flight, we reject the first solution. Furthermore, $v(r)$ will remain positive if and only if $-2gR + v_0^2 \geq 0$ so $v_0 \geq \sqrt{2gR}$.

```
√2gR /. {g → .00609, R → 3960}
6.94498
```

The escape velocity is 6.94498 miles per second.

11.5 According to Newton's law of cooling, the temperature of an object changes at a rate proportional to the difference in temperature between the object and the outside medium. If an object whose temperature is 70° Fahrenheit is placed in a medium whose temperature is 20°, and is found to be 40° after 3 minutes, what will its temperature be after 6 minutes?

SOLUTION

If $u(t)$ represents the temperature of the object at time t, $\dfrac{du}{dt} = k(u - 20)$. The initial condition is $u(0) = 70$.

```
DSolve[{u'[t] == k(u[t] - 20), u[0] == 70}, u[t], t]
{{u[t] → 20 + 50 e^(k t)}}
u[t_] = %[[1, 1, 2]]
20 + 50 e^(k t)
```

We determine k using the information about the temperature 3 minutes later. Since we are using `Solve` for a transcendental function e^x, *Mathematica* supplies a warning which may be safely ignored.

```
Solve[u[3] == 40, k]
Solve::ifun: Inverse functions are being
   used by Solve, so some solutions may not be found.
```

$\left\{\left\{k \to -\frac{1}{3}\text{Log}\left[\frac{5}{2}\right]\right\}\right\}$

```
u[6]/. k→%[[1, 1, 2]]
28
```

The temperature 6 minutes later is 28° Fahrenheit.

11.6 A freely falling body falls with an acceleration g, which is approximately 32.16 ft/sec^2. If air resistance is considered, its motion is changed dramatically. If an object whose mass is 5 slugs is dropped from a height of 1000 feet, determine how long it will take to hit the ground (a) neglecting air resistance and (b) assuming that the force of air resistance is equal to the velocity of the object.

SOLUTION

Let $h(t)$ represent the height of the object at time t, $v(t)$ its velocity, and $a(t)$ its acceleration. Recall that $v(t) = h'(t)$ and $a(t) = v'(t) = h''(t)$ and, by Newton's law, the sum of the external forces acting upon the object is equal to its mass times its acceleration: $m\,a(t) = \sum F$. In what follows we take "up" to be the positive direction.

(a) If air resistance is neglected, the only force acting on the object is gravity, so $m\,a(t) = -m\,g$. We can divide by m and solve the differential equation $h''(t) = -g$ with initial conditions $h'(0) = 0$, $h(0) = 1000$.

```
g = 32.16;
solution = DSolve[{h''[t] == -g, h'[0] == 0, h[0] == 1000}, h[t], t];
height[t_] = solution[[1, 1, 2]]
1000.-16.08t²
```

When the object reaches the ground its height will be 0.

```
Solve[height[t] == 0, t]
{{t→-7.886}, {t→7.886}}
```

It takes 7.886 seconds to reach the ground.

(b) If air resistance is taken into account, there is an external force acting upon the object, in addition to gravity, equal to $-v(t)$. The differential equation becomes

$$m\,a(t) = -m\,g - v(t)$$

or

$$m\,h''(t) = -m\,g - h'(t)$$

with initial conditions as in (a).

```
m = 5; g = 32.16;
solution = DSolve[{m h''[t] == -m g - h'[t], h'[0] == 0, h[0] == 1000}, h[t], t];
height[t_] = solution[[1, 1, 2]]
1804. - 804. e^(-0.2 t) - 160.8 t
FindRoot[height[t] == 0, {t, 10}]
{t→10.6213}
```

It now takes 10.6213 seconds to reach the ground.